********************************************************************************
/*REPLICATION FILES FOR:
"Belief in Divine (Versus Human) Control of Earth Affects Perceived Threat of Climate Change"
by John V. Kane and Samuel L. Perry.  Published in npj Climate Action
******************************************************************************** */


ssc install fre // install fre package
ssc install schemepack, replace // install additional schemes
ssc install blindschemes, replace  //  install additional schemes
ssc install g538schemes, replace //  install additional schemes
ssc install cibar // install cibar package
ssc install combomarginsplot // install combomarginsplot package
net install grc1leg, from("http://www.stata.com/users/vwiggins") replace // install grc1leg package

********************************************************
**#PRRI SURVEY DATA
********************************************************
use "Kane_Perry_PRRI_Survey.dta", clear

**#Original Survey Items Used

fre reli1 // religion
fre ppethm // race/ethnicity
fre party7_m // party
fre ideo_m // ideological self-placement
fre sex // gender identification
fre educ // education level
fre income // household income
tab age // respondent age
fre Q6_f // stewardship beliefs
fre Q8_e // dominion beliefs
fre relimp // importance of religion in one's life
fre Q2 // climate changed caused by human activity
fre Q3_b // climate change concern
fre Q4_a  // God would not allow humans to destroy the Earth
fre Q5_a Q5_b Q5_c Q5_d Q5_e Q5_f // climate change mitigation policies


**#Models & Figures with Cleaned/Recoded/Rescaled Variables

*Note:  some minor additional edits were made manually using "Graph Editor"

*Panel a of Figure 1
reg ClimateChangeProblem01 GodControl_01 i.reli1 revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age

margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// 
recastci(rarea) /// 
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// 
scheme(plottig) /// 
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(,  glcolor(gs15) glpattern(solid)) ///
ytitle("Climate Change Concern", size(small)) /// set y-title
xtitle("") ///
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// Leave (auto-generated) title blank (in other words, omit the title)
saving(g1_PRRI, replace)

*Panel b of Figure 1

logit ClimateChangeMajorProblem_YN GodControl_01 revrelimp_clean i.reli1 i.ppethm pid7 revideo_m_clean Female educ income age
margins, dydx(GodControl_01)
margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// how to specify the line color for first line in graph
recastci(rarea) /// fill in CI area with color
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// fill CI at 50% color, set color of line surrounding the CI
scheme(plottig) /// set scheme
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.54(.02).66,  glcolor(gs15) glpattern(solid)) ///
xtitle("") ///
ytitle("Pr(Climate Change is Major Concern/Crisis)", size(*.9)) /// set y-title
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// Leave (auto-generated) title blank (in other words, omit the title)
saving(g2_PRRI, replace)

*Panel c of Figure 1
logit HumanCaused GodControl_01 revrelimp_clean i.reli1 i.ppethm pid7 revideo_m_clean Female educ income age
margins, dydx(GodControl_01)
margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// how to specify the line color for first line in graph
recastci(rarea) /// fill in CI area with color
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// fill CI at 50% color, set color of line surrounding the CI
scheme(plottig) /// set scheme
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.5(.02).64,  glcolor(gs15) glpattern(solid)) ///
ytitle("Pr(Climate Change is Mostly Caused by Humans)", size(*.85)) /// set y-title
xtitle("{bf: Statement: God Would Not Allow Humans to Destroy the Earth}", size(vsmall) margin(top)) ///
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// Leave (auto-generated) title blank (in other words, omit the title)
saving(g3_PRRI, replace)


*Panel d of Figure 1

reg ClimatePolicyScale_01 GodControl_01 i.reli1 revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age

margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// how to specify the line color for first line in graph
recastci(rarea) /// fill in CI area with color
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// fill CI at 50% color, set color of line surrounding the CI
scheme(plottig) /// set scheme
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.53(.01).58,  glcolor(gs15) glpattern(solid)) ///
ytitle("Climate Policy Scale (0-1)", size(small)) /// set y-title
xtitle("{bf: Statement: God Would Not Allow Humans to Destroy the Earth}", size(vsmall) margin(top)) ///
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// Leave (auto-generated) title blank (in other words, omit the title)
saving(g4_PRRI, replace)

*Combined graph
graph combine "g1_PRRI.gph" "g2_PRRI.gph" "g3_PRRI.gph" "g4_PRRI.gph", ///
iscale(.8) scheme(white_tableau) xcommon xsize(6.5) ysize(5.5) ///
graphregion(margin(small)) ///
imargin(medsmall) ///
saving(g1234combined_PRRI, replace) //



********************************************************
**#SURVEY EXPERIMENT
********************************************************

use "Kane_Perry_SurveyExperiment.dta", clear

*Figure 2 (panel a)
cibar religexp_SMC1 if ReligionExp_clean2!=1, over(ReligionExp_clean2) /// 
barlabel(on) blfmt(%4.1f) blsize(medsmall) blposition(swest) blcolor(white) ///
level(95) /// 
ciopts(lcolor(red) lwidth(medium) type(rspike)) /// 
barcolor(edkblue%95 stgreen%95) /// 
baropts(lcolor(none) lwidth(medthin) fintensity(100)) /// 
bargap(100) /// 
graphopts( /// 
ytitle(Agree God Ultimately Has Control, col(gs8) size(medium)) /// 
ylab(1.8(.2)3.6, glcolor(white) glpattern(solid)) /// 
xlab(, nogrid) /// 
yscale(lcolor(black) lwidth(medium)) /// 
xscale(lcolor(black) lwidth(medium)) /// 
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) /// 
scheme(plottig)  /// 
legend(pos(6) ring(1) col(2) size(medium)) /// 
legend(title("Experimental Condition", size(medium) margin(tiny) box fcolor(gs8) color(white) bexpand)) /// 
) // 

graph save "Graph" "religSMC1.gph", replace

*Figure 2 (panel b)
cibar religexp_SMC2 if ReligionExp_clean2!=1, over(ReligionExp_clean2) /// 
barlabel(on) blfmt(%4.1f) blsize(medsmall) blposition(swest) blcolor(white) ///
level(95) /// 
ciopts(lcolor(red) lwidth(medium) type(rspike)) /// 
barcolor(edkblue%95 stgreen%95) /// 
baropts(lcolor(none) fintensity(100)) /// 
bargap(100) /// 
graphopts( /// 
ytitle(Agree God Would Not Allow Human Impact, col(gs8) size(medium)) /// titles y-axis
ylab(1.8(.2)3.6, glcolor(white) glpattern(solid)) /// 
xlab(, nogrid) ///
yscale(lcolor(black) lwidth(medium)) /// 
xscale(lcolor(black) lwidth(medium)) /// 
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) /// 
scheme(plottig)  /// 
legend(off) /// 
) // Note this last parenthesis! closes graphopts

graph save "Graph" "religSMC2.gph", replace

*Combined Figure
grc1leg "religSMC1.gph" "religSMC2.gph", ///
legendfrom("religSMC1.gph") ///
pos(6) ///
col(2) ///
ycommon ///
xcommon ///
iscale(.7) ///
graphregion(margin(small)) ///
xsize(6.5) ysize(5.5) ///
saving(combinedcibar.gph, replace)


*Figure 3 (panel a)
reg revreligexp_dv1_01 i.ReligionExp_clean if ReligionExp_clean!=1
estimates store religDV1

logit revreligexp_dv1_binary2 i.ReligionExp_clean if ReligionExp_clean!=1
margins, dydx(i.ReligionExp_clean) post
estimates store religDV1binary

reg religexp_dv2_01 i.ReligionExp_clean if ReligionExp_clean!=1
estimates store religDV2

logit revreligexp_dv6_binary i.ReligionExp_clean if ReligionExp_clean!=1
margins, dydx(i.ReligionExp_clean) post
estimates store religDV6


coefplot religDV1, drop(_cons) xline(0, lcolor(gs8) lwidth(medium)) scheme(plottig)  ///
xtitle(" ") /// 
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// 
xlab(-.08(.02).02, glpattern(solid) glcolor(white) nolabel notick gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge) mcolor(stc3) mlcolor(edkblue) mlwidth(thin) ///
levels(95 90) ciopts(lcolor(edkblue stgreen) lwidth(medthick thick) recast(rspike rcap)) ///
subtitle("Climate Change" "Concern", size(medsmall) color(white) bexpand box fcolor(edkblue) position(9) ///
orientation(vertical) margin(vsmall)) ///
saving(religdv1, replace)

*Figure 3 (panel b)
coefplot religDV1binary, drop(_cons) xline(0, lcolor(gs8) lwidth(medium)) scheme(plottig)  ///
xtitle(" ") /// 
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// 
xlab(-.08(.02).02, glpattern(solid) glcolor(white) nolabel notick gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge) mcolor(stc3) mlcolor(edkblue) mlwidth(thin) ///
levels(95 90) ciopts(lcolor(edkblue stgreen) lwidth(medthick thick) recast(rspike rcap)) ///
subtitle("Climate Change is a" "Maj.Problem/Crisis", size(medsmall) color(white) bexpand box fcolor(edkblue) position(9) ///
orientation(vertical) margin(vsmall)) ///
saving(religdv1b, replace)

*Figure 3 (panel c)
coefplot religDV2, ///
drop(_cons) xline(0, lcolor(gs8) lwidth(medium)) scheme(plottig)  ///
xtitle(" ") /// 
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// 
xlab(-.08(.02).02, glpattern(solid) glcolor(white) nolabel notick gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge) mcolor(stc3) mlcolor(edkblue) mlwidth(thin) ///
levels(95 90) ciopts(lcolor(edkblue stgreen) lwidth(medthick thick) recast(rspike rcap)) ///
subtitle("Climate Change" "Needs Action", size(medsmall) color(white) bexpand box fcolor(edkblue) position(9) ///
orientation(vertical) margin(vsmall)) ///
saving(religdv2, replace)

*Figure 3 (panel d)
coefplot religDV6, drop(_cons) xline(0, lcolor(gs8) lwidth(medium)) scheme(plottig)  ///
xtitle("{bf: Effect of 'God is in Control' Treatment}", size(medsmall) margin(medium)) /// 
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// 
xlab(-.08(.02).02, glpattern(solid) glcolor(white) gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge) mcolor(stc3) mlcolor(edkblue) mlwidth(thin) ///
levels(95 90) ciopts(lcolor(edkblue stgreen) lwidth(medthick thick) recast(rspike rcap)) ///
subtitle("Pr(Request" "NOAA Info)", size(medsmall) color(white) bexpand box fcolor(edkblue) position(9) ///
orientation(vertical) margin(vsmall)) ///
saving(religdv6, replace)

*Combined figure
graph combine "religdv1.gph" "religdv1b.gph" "religdv2.gph" "religdv6.gph", ///
col(1) ///
ycommon ///
xcommon ///
iscale(.6) ///
imargin(vsmall) ///
graphregion(margin(small)) ///
xsize(6.5) ysize(5.5) /// 
saving(combinedgraph_religDV126, replace)


********************************************************
**#SUPPLEMENTAL ANALYSES (PRRI DATA)
********************************************************

use "Kane_Perry_SurveyExperiment.dta", clear

**SUPPLEMENTAL FIGURE 1-1
graph bar,  /// basic command for bar graph with counts
over(GodControl, gap(*.9)) /// identify the "over" variable; specify distance between bars (here)
blabel(bar, color(white) format(%3.0f) size(medsmall) pos(inside)) /// format bar labels
ylab(, glcolor(gs15) glstyle(solid)) /// add light gray, solid horizontal lines
bar(1, fcolor(stgreen%85) fintensity(inten100) lcolor(none)) /// format look of bars
scheme(plottig) /// specify scheme
b1title("God Would Not Let Humans Destroy Earth (Agreement Level)") /// x-axis title
ytitle(Percent) /// title y-axis
graphregion(margin(vsmall)) /// small margin between plot and outer edge of graph
xsize(6.5) ysize(4.5) // specify graph dimensions in inches

**SUPPLEMENTAL FIGURE 1-2
graph bar, over(GodControl, gap(*.2))  ///
by(relig6cat, total col(2) note("") ///
subtitle(, box color(gs4) fcolor(gs15) lcolor(midblue) lwidth(vthin)  bexpand)) /// 
blabel(bar, color(white) format(%4.0f) size(small) pos(inside) alignment(top)) /// 
asyvars ///
legend(pos(6) row(1)) /// 
ylab(, glcolor(gs15) glstyle(solid)) /// 
bar(1, fcolor(edkblue%95) fintensity(inten100) lcolor(none) ) /// 
bar(2, fcolor(stgreen%95) fintensity(inten100) lcolor(none) ) /// 
bar(3, fcolor(stc9%95) fintensity(inten100) lcolor(none) ) /// 
bar(4, fcolor(stc4%95) fintensity(inten100) lcolor(none) ) /// 
scheme(plottig) /// 
ytitle(Percent of Group) /// 
graphregion(margin(vsmall)) /// 
legend(title("God Would Not Allow Humans to Destroy Earth (Agreement Level)", size(medsmall) margin(tiny) box fcolor(gs8) color(white) bexpand)) /// 
xsize(6.5) ysize(4.5) // 


**SUPPLEMENTAL FIGURE 1-3
cibar GodControl, over(relig6cat) ///
barlabel(on) blfmt(%4.1f) blsize(medsmall) blposition(swest) blcolor(white) ///
level(83) /// specifies the CI level to be 95% (this is the default)
ciopts(lcolor(red) lwidth(medium) type(rspike)) /// changes CI color, thickness, and style
barcolor(edkblue%95 stgreen%95 stc9%95 stc4%95) /// colors each bar in order of the bars
baropts(lcolor(none) lwidth(medthin) fintensity(100)) /// changes the bar outlines and color intensity
bargap(100) /// places a big gap between the bars (otherwise there will be no gap)
graphopts( /// start of graphopts option
ytitle(Agree God Would Not Let Humans Destroy Earth, col(gs8) size(medium)) /// titles y-axis
ylab(1(.1)2.4, glcolor(white) glpattern(solid)) /// y-axis label options and horizontal gridline options
xlab(, nogrid) /// x-axis label options and vertical gridline options
yscale(lcolor(black) lwidth(medium)) /// change y-axis line look
xscale(lcolor(black) lwidth(medium)) /// change x-axis line look
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) /// graph dimensions and margin b/w plot and edge 
scheme(plottig)  /// use stcolor scheme (Stata 18 only)
legend(pos(6) ring(1) row(2) size(medsmall)) /// legend options
legend(title("Religious Group", size(medium) margin(tiny) box fcolor(gs8) color(white) bexpand)) /// legend title
) // Note this last parenthesis! closes graphopts


*SUPPLEMENTAL FIGURE 1-4
cibar GodControl, over(revrelimp_clean) ///
barlabel(on) blfmt(%4.1f) blsize(medsmall) blposition(swest) blcolor(white) ///
level(83) /// 
ciopts(lcolor(red) lwidth(medium) type(rspike)) /// 
barcolor(edkblue%95 stgreen%95 stc9%95 stc4%95) /// 
baropts(lcolor(none) lwidth(medthin) fintensity(100)) /// 
bargap(100) /// 
graphopts( /// 
ytitle(Agree God Would Not Let Humans Destroy Earth, col(gs8) size(medium)) /// 
ylab(1(.1)2.4, glcolor(white) glpattern(solid)) /// 
xlab(, nogrid) /// 
yscale(lcolor(black) lwidth(medium)) /// 
xscale(lcolor(black) lwidth(medium)) /// 
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) /// 
scheme(plottig)  /// 
legend(pos(6) ring(1) row(2) size(medsmall)) /// 
legend(title("Importance of Religion in Respondent's Life", size(medium) margin(tiny) box fcolor(gs8) color(white) bexpand)))

*SUPPLEMENTAL TABLE 1-1
pwcorr revQ5_a_clean revQ5_b_clean revQ5_c_clean revQ5_d_clean revQ5_e_clean revQ5_f_clean, star(.05)

*Cronbach's alpha measures
alpha revQ5_a_clean revQ5_b_clean revQ5_c_clean revQ5_d_clean revQ5_f_clean, item //  missing item e (only asked of half the sample)
alpha revQ5_a_clean revQ5_b_clean revQ5_c_clean revQ5_e_clean revQ5_f_clean, item // missing item d (only asked of half the sample)


*SUPPLEMENTAL TABLE 2-2
tab GodControl revdominion, col nofreq chi2 V

*SUPPLEMENTAL TABLE 2-3
tab GodControl revstewardship, col nofreq chi2 V

*SUPPLEMENTAL TABLE 2-4
pwcorr GodControl revstewardship revdominion, star(.05)


*SUPPLEMENTAL TABLE 2-5
*Model 1 (Concern)
reg ClimateChangeProblem01 GodControl_01 revdominion01 revstewardship01 i.reli1 revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age

*Model 2 (Major Problem)
logit ClimateChangeMajorProblem_YN GodControl_01 revdominion01 revstewardship01 revrelimp_clean i.reli1 i.ppethm pid7 revideo_m_clean Female educ income age
margins, dydx(GodControl_01)


*Model 3 (Human Cause)
logit HumanCaused GodControl_01 revdominion01 revstewardship01 revrelimp_clean i.reli1 i.ppethm pid7 revideo_m_clean Female educ income age
margins, dydx(GodControl_01)

*Model 4 (Policy Scale)
reg ClimatePolicyScale_01 GodControl_01 revdominion01 revstewardship01 i.reli1 revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age


*SUPPLEMENTAL FIGURE 2-1
*Top-left panel (Concern)
reg ClimateChangeProblem01 GodControl_01  revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age if born==1
margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// how to specify the line color for first line in graph
recastci(rarea) /// fill in CI area with color
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// fill CI at 50% color, set color of line surrounding the CI
scheme(plottig) /// set scheme
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.36(.02).48,  glcolor(gs15) glpattern(solid)) ///
ytitle("Climate Change Concern", size(small)) /// set y-title
xtitle("") ///
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// Leave (auto-generated) title blank (in other words, omit the title)
saving(g1_PRRI_Evang, replace)

*Top-right panel (Major problem)
logit ClimateChangeMajorProblem_YN GodControl_01 revrelimp_clean  i.ppethm pid7 revideo_m_clean Female educ income age if born==1
margins, dydx(GodControl_01)
margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// how to specify the line color for first line in graph
recastci(rarea) /// fill in CI area with color
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// fill CI at 50% color, set color of line surrounding the CI
scheme(plottig) /// set scheme
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.39(.04).59,  glcolor(gs15) glpattern(solid)) ///
xtitle("") ///
ytitle("Pr(Climate Change is Major Concern/Crisis)", size(*.9)) /// set y-title
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// Leave (auto-generated) title blank (in other words, omit the title)
saving(g2_PRRI_Evang, replace)

*Bottom-left panel (Caused by humans)
logit HumanCaused GodControl_01 revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age if born==1
margins, dydx(GodControl_01)
margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// 
recastci(rarea) /// 
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// 
scheme(plottig) /// 
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.34(.04).46,  glcolor(gs15) glpattern(solid)) ///
ytitle("Pr(Climate Change is Mostly Caused by Humans)", size(*.85)) /// set y-title
xtitle("{bf: Statement: God Would Not Allow Humans to Destroy the Earth}", size(vsmall) margin(top)) ///
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// 
saving(g3_PRRI_Evang, replace)

*Bottom-right panel (Climate policy scale)
*Climate policy scale Model
reg ClimatePolicyScale_01 GodControl_01 revrelimp_clean i.ppethm pid7 revideo_m_clean Female educ income age if born==1

margins, at(GodControl_01=(0(.33)1)) // 

marginsplot, ///
plotopts(lcolor(stc3) lwidth(medthick) msize(*1.5) mcolor(white) mlcolor(edkblue) mlwidth(thick)) /// how to specify the line color for first line in graph
recastci(rarea) /// fill in CI area with color
ciopts(fcolor(stgreen%50) lcolor(stc3) lwidth(medthin)) /// fill CI at 50% color, set color of line surrounding the CI
scheme(plottig) /// 
xlab(, labsize(vsmall) glcolor(gs15) glpattern(solid)) ///
ylab(.40(.02).48,  glcolor(gs15) glpattern(solid)) ///
ytitle("Climate Policy Scale (0-1)", size(small)) /// set y-title
xtitle("{bf: Statement: God Would Not Allow Humans to Destroy the Earth}", size(vsmall) margin(top)) ///
xlabel(0 `""Completely" "Disagree""' .3333 `""Mostly" "Disagree""' .6666 `""Mostly" "Agree""' 1 `""Completely" "Agree""') ///
xsize(6.5) ysize(4.5) graphregion(margin(vsmall)) ///
xscale(lcolor(gs8) lwidth(medium) line) ///
yscale(lcolor(gs8) lwidth(medium) line) ///
graphregion(margin(medium)) ///
title("")  /// 
saving(g4_PRRI_Evang, replace)


*Combined graph
graph combine "g1_PRRI_Evang.gph" "g2_PRRI_Evang.gph" "g3_PRRI_Evang.gph" "g4_PRRI_Evang.gph", ///
iscale(.8) scheme(white_tableau) xcommon xsize(6.5) ysize(5.5) ///
graphregion(margin(small)) ///
imargin(medsmall) ///
saving(g1234combined_PRRI, replace) //


********************************************************
**#SUPPLEMENTAL ANALYSES (SURVEY EXPERIMENT DATA)
********************************************************

use "Kane_Perry_SurveyExperiment.dta", clear


*Figure 3-1
graph bar,  /// basic command for bar graph with counts
over(revreligexp_dv1, gap(*.9)) /// identify the "over" variable; specify distance between bars (here)
blabel(bar, color(white) format(%3.0f) size(medsmall) pos(inside)) /// format bar labels
ylab(, glcolor(gs15) glstyle(solid)) /// add light gray, solid horizontal lines
bar(1, fcolor(stgreen%85) fintensity(inten100) lcolor(none)) /// format look of bars
scheme(plottig) /// specify scheme
b1title("Outcome: Concern") /// x-axis title
ytitle(Percent) /// title y-axis
graphregion(margin(vsmall)) /// small margin between plot and outer edge of graph
xsize(6.5) ysize(4.5) // specify graph dimensions in inches

*Figure 3-2

graph bar,  /// basic command for bar graph with counts
over(revreligexp_dv1_binary2, gap(*.9)) /// identify the "over" variable; specify distance between bars (here)
blabel(bar, color(white) format(%3.0f) size(medsmall) pos(inside)) /// format bar labels
ylab(, glcolor(gs15) glstyle(solid)) /// add light gray, solid horizontal lines
bar(1, fcolor(stgreen%85) fintensity(inten100) lcolor(none)) /// format look of bars
scheme(plottig) /// specify scheme
b1title("Outcome: Crisis") /// x-axis title
ytitle(Percent) /// title y-axis
graphregion(margin(vsmall)) /// small margin between plot and outer edge of graph
xsize(6.5) ysize(4.5) // specify graph dimensions in inches

*Figure 3-3
graph bar,  /// basic command for bar graph with counts
over(revreligexp_dv1_binary2, gap(*.9)) /// identify the "over" variable; specify distance between bars (here)
blabel(bar, color(white) format(%3.0f) size(medsmall) pos(inside)) /// format bar labels
ylab(, glcolor(gs15) glstyle(solid)) /// add light gray, solid horizontal lines
bar(1, fcolor(stgreen%85) fintensity(inten100) lcolor(none)) /// format look of bars
scheme(plottig) /// specify scheme
b1title("Outcome: Crisis") /// x-axis title
ytitle(Percent) /// title y-axis
graphregion(margin(vsmall)) /// small margin between plot and outer edge of graph
xsize(6.5) ysize(4.5) // specify graph dimensions in inches

*Figure 3-4
graph bar,  /// basic command for bar graph with counts
over(revreligexp_dv6_binary, gap(*.9)) /// identify the "over" variable; specify distance between bars (here)
blabel(bar, color(white) format(%3.0f) size(medsmall) pos(inside)) /// format bar labels
ylab(, glcolor(gs15) glstyle(solid)) /// add light gray, solid horizontal lines
bar(1, fcolor(stgreen%85) fintensity(inten100) lcolor(none)) /// format look of bars
scheme(plottig) /// specify scheme
b1title("Outcome: Action") /// x-axis title
ytitle(Percent) /// title y-axis
graphregion(margin(vsmall)) /// small margin between plot and outer edge of graph
xsize(6.5) ysize(4.5) // specify graph dimensions in inches

*Table 4-1
tab religexp_FMC ReligionExp_clean if ReligionExp_clean!=1, col chi2 V // 

*Table 4-2
	*Model 1
reg religexp_SMC1 i.ReligionExp_clean if ReligionExp_clean!=1
	*Model 2
reg religexp_SMC2 i.ReligionExp_clean if ReligionExp_clean!=1

*Table 4-4
	*Model 1
reg revreligexp_dv3_01 i.ReligionExp_clean if ReligionExp_clean!=1 // this one is policy-related but not explicit climate-change; non-sig

	*Model 2
reg revreligexp_dv4_01 i.ReligionExp_clean if ReligionExp_clean!=1 // this one is policy-related; non-sig
	
	*Model 3
reg revreligexp_dv5_01 i.ReligionExp_clean if ReligionExp_clean!=1 // this one is candidate-related; non-sig


*Table 5-1
*Model 1  
reg revreligexp_dv1_01 i.ReligionExp_clean i.gender age i.religion_pew i.race famine_ces education ///
i.D0I1R2 ideo_clean ReligiosityScale01 if ReligionExp_clean!=1

*Model 2  
logit revreligexp_dv1_binary2 i.ReligionExp_clean i.gender age i.religion_pew i.race famine_ces education ///
i.D0I1R2 ideo_clean ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) 

*Model 3  
reg religexp_dv2_01 i.ReligionExp_clean i.gender age i.religion_pew i.race famine_ces education ///
i.D0I1R2 ideo_clean ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) 

*Model 4  
logit revreligexp_dv6_binary i.ReligionExp_clean i.gender age i.religion_pew i.race famine_ces education ///
i.D0I1R2 ideo_clean ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) 


*Figure 5-1
*Concern
reg revreligexp_dv1_01 i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV1att, replace)

*Major problem
logit revreligexp_dv1_binary2 i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV1binaryatt, replace)

*Needs action
reg religexp_dv2_01 i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV2att, replace)

*Request NOAA info
logit revreligexp_dv6_binary i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV6att, replace)

*Combined figure
combomarginsplot religDV1att religDV1binaryatt religDV2att religDV6att, scheme(plottig) ///
labels("Concern About Climate Change" "Climate Change is a Maj.Problem/Crisis" "Climate Change Needs Action" "Request NOAA Info") ///
ytitle("{bf:Effect on Outcome (percentage points)}") ///
xsize(6.5) ysize(5.5) ///
file1opts(lcolor(navy%90) lwidth(medthick) msize(medlarge) mcolor(black) mlcolor(navy) mlwidth(medthick)) ///
file2opts(lcolor(stblue%90) lwidth(medthick) msize(medlarge) msymbol(triangle) mcolor(black) mlcolor(stblue) mlwidth(medthick)) ///
file3opts(lcolor(stgreen%90) lwidth(medthick) msize(medlarge) msymbol(square) mcolor(black) mlcolor(stgreen) mlwidth(medthick)) ///
file4opts(lcolor(stc5%90) lwidth(medthick) msize(medlarge) msymbol(square) mcolor(black) mlcolor(stc5) mlwidth(medthick)) ///
fileci1opts(lwidth(medthick) lcolor(navy%90) recast(rspike)) ///
fileci2opts(lwidth(medthick) lcolor(stblue%90) recast(rspike)) ///
fileci3opts(lwidth(medthick) lcolor(stgreen%90) recast(rspike)) ///
fileci4opts(lwidth(medthick) lcolor(stc5%90) recast(rspike)) ///
graphregion(margin(medium)) ///
xlab(, glpattern(solid) glcolor(white)) ///
ylab(, glpattern(solid) glcolor(white)) ///
offset(.10) ///
xtitle("Pre-Treatment Attentiveness") ///
yline(0) ///
legend (ring(0) pos(1) col(1) size(vsmall)) title("") ///
saving(combomargins_religdvsatt.gph, replace)


*Figure 5-2
*Renewables
reg religexp_dv3_01 i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1 // 
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV3att, replace)

*Emissions
reg religexp_dv4_01 i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1 // 
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV4att, replace)

*Candidate
reg religexp_dv5_01 i.ReligionExp_clean##c.mvcscale if ReligionExp_clean!=1 //
margins, dydx(i.ReligionExp_clean) at(mvcscale=(0(1)3)) saving(religDV5att, replace)

combomarginsplot religDV3att religDV4att religDV5att, scheme(plottig) ///
labels("Renewables" "Emissions" "Candidate") ///
ytitle("{bf:Effect on Outcome (percentage points)}") ///
xsize(6.5) ysize(5.5) ///
file1opts(lcolor(navy%90) lwidth(medthick) msize(medlarge) mcolor(black) mlcolor(navy) mlwidth(medthick)) ///
file2opts(lcolor(stblue%90) lwidth(medthick) msize(medlarge) msymbol(triangle) mcolor(black) mlcolor(stblue) mlwidth(medthick)) ///
file3opts(lcolor(stgreen%90) lwidth(medthick) msize(medlarge) msymbol(square) mcolor(black) mlcolor(stgreen) mlwidth(medthick)) ///
fileci1opts(lwidth(medthick) lcolor(navy%90) recast(rspike)) ///
fileci2opts(lwidth(medthick) lcolor(stblue%90) recast(rspike)) ///
fileci3opts(lwidth(medthick) lcolor(stgreen%90) recast(rspike)) ///
graphregion(margin(medium)) ///
xlab(, glpattern(solid) glcolor(white)) ///
ylab(, glpattern(solid) glcolor(white)) ///
offset(.10) ///
xtitle("Pre-Treatment Attentiveness") ///
yline(0) ///
legend (ring(0) pos(1) col(1) size(small)) title("") ///
saving(combomargins_religdvs_policy_att.gph, replace)


*Table 6-1
*Model 1
reg revreligexp_dv1_01 b1.ReligionExp_clean // mostly human control

*Model 2
logit revreligexp_dv1_binary2 b1.ReligionExp_clean
margins, at(ReligionExp_clean=(0(1)2)) //

*Model 3
reg religexp_dv2_01 b1.ReligionExp_clean // about equal contributions

*Model 4
logit revreligexp_dv6_binary b1.ReligionExp_clean 
margins, at(ReligionExp_clean=(0(1)2)) //


*Table 6-2
*Model 1
reg revreligexp_dv1_01 i.ReligionExp_clean##i.BornAgain_Evangelical if ReligionExp_clean!=1

*Model 2
logit revreligexp_dv1_binary2 i.ReligionExp_clean##i.BornAgain_Evangelical if ReligionExp_clean!=1

*Model 3
reg religexp_dv2_01 i.ReligionExp_clean##i.BornAgain_Evangelical if ReligionExp_clean!=1

*Model 4
logit revreligexp_dv6_binary i.ReligionExp_clean##i.BornAgain_Evangelical if ReligionExp_clean!=1



*Figure 6-1
*Top-left panel
reg revreligexp_dv1_01 i.ReligionExp_clean##c.ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) at(ReligiosityScale01=(0(.1)1))
marginsplot, yline(0) ///
xlab(, glpattern(solid) glcolor(gs15)) ///
ylab(, glpattern(solid) glcolor(gs15)) ///
plotopts(mcolor(stgreen) msize(medium) mlcolor(dkgreen) lcolor(stgreen)) ///
ciopts(lwidth(medthick) lcolor(stgreen) recast(rspike)) ///
title(" ") ///
xtitle(Religiosity Scale) ///
ytitle("Effect on Concern") ///
saving(margins_religinteraction1.gph, replace)

*Top-right panel
logit revreligexp_dv1_binary2 i.ReligionExp_clean##c.ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) at(ReligiosityScale01=(0(.1)1))
marginsplot, yline(0) ///
xlab(, glpattern(solid) glcolor(gs15)) ///
ylab(, glpattern(solid) glcolor(gs15)) ///
plotopts(mcolor(stgreen) msize(medium) mlcolor(dkgreen) lcolor(stgreen)) ///
ciopts(lwidth(medthick) lcolor(stgreen) recast(rspike)) ///
title(" ") ///
xtitle(Religiosity Scale) ///
ytitle("Effect on pr(Maj.Prob/Crisis)") ///
saving(margins_religinteraction2.gph, replace)


*Bottom-left panel
reg religexp_dv2_01 i.ReligionExp_clean##c.ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) at(ReligiosityScale01=(0(.1)1))
marginsplot, yline(0) ///
xlab(, glpattern(solid) glcolor(gs15)) ///
ylab(, glpattern(solid) glcolor(gs15)) ///
plotopts(mcolor(stgreen) msize(medium) mlcolor(dkgreen) lcolor(stgreen)) ///
ciopts(lwidth(medthick) lcolor(stgreen) recast(rspike)) ///
title(" ") ///
xtitle(Religiosity Scale) ///
ytitle("Effect on Action") ///
saving(margins_religinteraction3.gph, replace)

*Bottom-right panel
logit revreligexp_dv6_binary i.ReligionExp_clean##c.ReligiosityScale01 if ReligionExp_clean!=1
margins, dydx(ReligionExp_clean) at(ReligiosityScale01=(0(.1)1))
marginsplot, yline(0) ///
xlab(, glpattern(solid) glcolor(gs15)) ///
ylab(, glpattern(solid) glcolor(gs15)) ///
plotopts(mcolor(stgreen) msize(medium) mlcolor(dkgreen) lcolor(stgreen)) ///
ciopts(lwidth(medthick) lcolor(stgreen) recast(rspike)) ///
title(" ") ///
xtitle(Religiosity Scale) ///
ytitle("Effect on pr(Request NOAA Information)") ///
saving(margins_religinteraction4.gph, replace)

*Combined graph
graph combine "margins_religinteraction1.gph" "margins_religinteraction2.gph" "margins_religinteraction3.gph" ///
"margins_religinteraction4.gph", ///
xcommon ///
iscale(.7) ///
imargin(vsmall) ///
xsize(6.5) ysize(5.5) graphregion(margin(vsmall))


*Figure 6-2

*Top panel estimates
reg revreligexp_dv1_01 i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==0
estimates store religDV1D
reg revreligexp_dv1_01 i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==1
estimates store religDV1R

*Second panel estimates
reg revreligexp_dv1_binary2 i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==0
estimates store religDV1Db
reg revreligexp_dv1_binary i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==1
estimates store religDV1Rb

*Third panel estimates
reg religexp_dv2_01 i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==0
estimates store religDV2D
reg religexp_dv2_01 i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==1
estimates store religDV2R

*Fourth panel estimates
logit revreligexp_dv6_binary i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==0
margins, dydx(i.ReligionExp_clean) post
estimates store religDV6D

logit revreligexp_dv6_binary i.ReligionExp_clean if ReligionExp_clean!=1 & D0R1==1
margins, dydx(i.ReligionExp_clean) post
estimates store religDV6R

	*Testing interactions
reg revreligexp_dv1_01 i.ReligionExp_clean##i.D0R1 if ReligionExp_clean!=1 
logit revreligexp_dv1_binary2 i.ReligionExp_clean##i.D0R1 if ReligionExp_clean!=1 
reg religexp_dv2_01 i.ReligionExp_clean##i.D0R1 if ReligionExp_clean!=1 
logit revreligexp_dv6_binary i.ReligionExp_clean##i.D0R1 if ReligionExp_clean!=1 


*Top panel
coefplot (religDV1D, label("Democrats") ///
levels(95 90) ciopts(lcolor(gs6 midblue) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(square) mcolor(midblue) mlcolor(cyan))  ///
(religDV1R, label("Republicans") ///
levels(95 90) ciopts(lcolor(gs6 red) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(circle) mcolor(red) mlcolor(cyan)), ///
drop(_cons) xline(0, lcolor(gs8)lwidth(medium)) scheme(white_jet) ///
xtitle(" ") /// bolded text
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// enlarge size of y-axis labels
xlab(-.14(.02) .06, glpattern(solid) glcolor(gs14) nolabel notick gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge)  mlwidth(medthin) ///
subtitle("Climate Change" "Concern", size(medsmall) color(white) bexpand box fcolor(black) position(9) ///
orientation(vertical) margin(small)) ///
legend(ring(1) pos(6) col(1) size(tiny) keygap(tiny) margin(tiny) box color(white) fcolor(black) region(fcolor(black))) ///
saving(religdv1_partisans, replace)

*Second panel
coefplot (religDV1Db, label("Democrats") ///
levels(95 90) ciopts(lcolor(gs6 midblue) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(square) mcolor(midblue) mlcolor(cyan))  ///
(religDV1Rb, label("Republicans") ///
levels(95 90) ciopts(lcolor(gs6 red) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(circle) mcolor(red) mlcolor(cyan)), ///
drop(_cons) xline(0, lcolor(gs8)lwidth(medium)) scheme(white_jet) ///
xtitle(" ", size(medsmall) margin(medium)) /// 
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// enlarge size of y-axis labels
xlab(-.14(.02) .06, notick nolabel glpattern(solid) glcolor(gs14) gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge)  mlwidth(medthin) ///
subtitle("Climate Change is a" "Maj.Problem/Crisis", size(medsmall) color(white) bexpand box fcolor(black) position(9) ///
orientation(vertical) margin(vsmall)) ///
legend(off) ///
saving(religdv1b_partisans, replace)

*Third panel
coefplot (religDV2D, label("Democrats") ///
levels(95 90) ciopts(lcolor(gs6 midblue) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(square) mcolor(midblue) mlcolor(cyan))  ///
(religDV2R, label("Republicans") ///
levels(95 90) ciopts(lcolor(gs6 red) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(circle) mcolor(red) mlcolor(cyan)), ///
drop(_cons) xline(0, lcolor(gs8)lwidth(medium)) scheme(white_jet) ///
xtitle(" ") /// bolded text
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// enlarge size of y-axis labels
xlab(-.14(.02) .06, glpattern(solid) nolabel notick glcolor(gs14) gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge)  mlwidth(medthin) ///
subtitle("Climate Change" "Needs Action", size(medsmall) color(white) bexpand box fcolor(black) position(9) ///
orientation(vertical) margin(vsmall)) ///
legend(off) ///
saving(religdv2_partisans, replace)

*Fourth panel
coefplot (religDV6D, label("Democrats") ///
levels(95 90) ciopts(lcolor(gs6 midblue) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(square) mcolor(midblue) mlcolor(cyan))  ///
(religDV6R, label("Republicans") ///
levels(95 90) ciopts(lcolor(gs6 red) lwidth(medthick thick) recast(rspike rcap)) ///
msymbol(circle) mcolor(red) mlcolor(cyan)), ///
drop(_cons) xline(0, lcolor(gs8)lwidth(medium)) scheme(white_jet) ///
xtitle("Effect of 'God is in Control' Treatment", size(medsmall) margin(medium)) /// 
graphregion(margin(medium)) ///
xsize(6.5) ysize(4.5) ///
grid(none) /// 
ylab(, labsize(*1.1) notick nolabel) /// enlarge size of y-axis labels
xlab(-.14(.02) .06, glpattern(solid) glcolor(gs14) gmin gmax gextend) ///
coeflabels(2.ReligionExp_clean="God in Control") ///
mlabel mlabcolor(white) mlabsize(small) format(%4.2f)  ///
mlabposition(center) msize(ehuge)  mlwidth(medthin) ///
subtitle("Pr(Request" "NOAA Info)", size(small) color(white) bexpand box fcolor(black) position(9) ///
orientation(vertical) margin(vsmall)) ///
legend(off) ///
saving(religdv6_partisans, replace)

*Combined panel
grc1leg "religdv1_partisans.gph" "religdv1b_partisans" "religdv2_partisans.gph" "religdv6_partisans.gph", ///
legendfrom("religdv1_partisans.gph") ///
pos(1) ring(0) ///
col(1) ///
ycommon ///
xcommon ///
iscale(.6) ///
graphregion(margin(small)) ///
xsize(6.5) ysize(5.5) //

*********************************
*********************************
